CAPÍTULO 6

APLICACIONES AL CÁLCULO

1.- CÁLCULO DE LÍMITES

El límite de una sucesión o de una función se calcula a través del comando **Limit**, cuya sintaxis es:

Limit[f,x->a]

In[1]:= $\text{Limit}\left[\frac{x^3 - 3x}{3x^3 - 2}, x \to \infty\right]$

Out[1]= $\frac{1}{3}$

In[4]:= $\text{Limit}\left[\left(\frac{2x}{3 + 2x}\right)^{5x}, x \to \infty\right]$

Out[4]= $\frac{1}{e^{15/2}}$

In[5]:= $\text{Limit}\left[\frac{1 - \text{Cos}[x]}{x^2}, x \to 0\right]$

Out[5]= $\frac{1}{2}$

El comando **Limit** permite calcular los límites laterales de una función ejecutando la instrucción:

Limit[f,x->a,Direction->b]

siendo b igual a 1 para calcular el límite por la izquierda y –1 para hallar el límite por la derecha.

In[7]:= Plot[Tan[x], {x, 0, π}];



In[8]:= $\text{Limit}\left[\text{Tan}[x], x \to \frac{\pi}{2}, \text{Direction} \to 1\right]$

Out[8]= ∞

In[9]:= $\text{Limit}\left[\text{Tan}[x], x \to \frac{\pi}{2}, \text{Direction} \to -1\right]$

Out[9]= $-\infty$

In[14]:= $\text{Limit}\left[(x + 3)^{\frac{2}{x}}, x \to 0, \text{Direction} \to 1\right]$

Out[14]= 0

In[15]:= $\text{Limit}\left[(x + 3)^{\frac{2}{x}}, x \to 0, \text{Direction} \to -1\right]$

Out[15]= ∞

2.- CÁLCULO DIFERENCIAL

La función derivada de una función f se obtiene a partir del comando **D**, cuya sintaxis es:

D[f,var]

siendo var la variable con respecto a la que hallar la derivada. También se puede utilizar el símbolo correspondiente que aparece en la paleta BasicInput.

```
In[16]:= D[(x - 1)/(x^2 + 1), x]
Out[16]= -(2 (-1 + x) x)/(1 + x^2)^2 + 1/(1 + x^2)

In[17]:= ∂x (x - 1)/(x^2 + 1)
Out[17]= -(2 (-1 + x) x)/(1 + x^2)^2 + 1/(1 + x^2)
```

```
In[21]:= D[√(a^2 - x^2) + a ArcSin[x/a], x]
Out[21]= -x/√(a^2 - x^2) + 1/√(1 - x^2/a^2)

In[23]:= ∂x Log[√(1 + e^x) - 1]
Out[23]= e^x/(2 √(1 + e^x) (-1 + √(1 + e^x)))

In[24]:= Simplify[%]
Out[24]= e^x/(2 + 2 e^x - 2 √(1 + e^x))

In[25]:= Simplify[∂x √((x + a) (x + b) (x + c))]
Out[25]= ((a + x) (b + x) + (a + x) (c + x) + (b + x) (c + x))/(2 √((a + x) (b + x) (c + x)))
```

El comando **D** se utiliza también, para hallar las funciones derivadas de orden superior, sin más que emplear la sintaxis:

D[f,{var,n}]

siendo n el orden de la función derivada que se desea obtener de la función f respecto a la variable var.

```
In[26]:= D[Sin[x]^3, {x, 4}]
Out[26]= -60 Cos[x]^2 Sin[x] + 21 Sin[x]^3

In[27]:= ∂{x,4} Sin[x]^3
Out[27]= -60 Cos[x]^2 Sin[x] + 21 Sin[x]^3

In[28]:= ∂{x,3} Log[3 x + 2 y]
Out[28]= 54/(3 x + 2 y)^3
```

Para una función con más de una variable, el comando **D** calcula las derivadas parciales con respecto a cada una de las variables que se indiquen. Además se pueden calcular las derivadas parciales de orden superior con respecto a cada una de las variables, utilizando las notaciones:

D[f,var1,var2,...]

D[f,{var1,n1},{var2,n2},...]

In[29]:= `D[Log[3 x + 2 y], x, y]`

Out[29]= $-\frac{6}{(3\,x+2\,y)^2}$

In[30]:= `D[Log[3 x + 2 y], {x, 2}, {y, 3}]`

Out[30]= $\frac{1728}{(3\,x+2\,y)^5}$

In[31]:= $\partial_{\{x,2\},\{y,3\}}$ `Log[3 x + 2 y]`

Out[31]= $\frac{1728}{(3\,x+2\,y)^5}$

In[32]:= $\texttt{D}\left[\texttt{Cos[x]}^{\texttt{Sin[x]}},\ \texttt{x}\right]\ \texttt{/. x} \rightarrow \frac{\pi}{4}$

Out[32]= $2^{-\frac{1}{2\sqrt{2}}}\left(-\frac{1}{\sqrt{2}}-\frac{\text{Log}[2]}{2\sqrt{2}}\right)$

In[33]:= `Simplify[%]`

Out[33]= $-2^{\frac{1}{4}\left(-6-\sqrt{2}\right)}\,(2+\text{Log}[2])$

Para derivar una función dada en forma implícita utilizaremos el comando **Dt**.

In[34]:= $\texttt{Dt}\left[x^2+y^2==4,\ x\right]$

Out[34]= $2\,x+2\,y\,\text{Dt}[y,\,x]==0$

In[35]:= `Solve[%, Dt[y, x]]`

Out[35]= $\left\{\left\{\text{Dt}[y,\,x]\rightarrow-\frac{x}{y}\right\}\right\}$

In[36]:= $\texttt{Dt}\left[e^{x+y}\left(x^2+y^2\right)==4,\ x\right]$

Out[36]= $e^{x+y}\,(x^2+y^2)\,(1+\text{Dt}[y,\,x])+e^{x+y}\,(2\,x+2\,y\,\text{Dt}[y,\,x])==0$

In[37]:= `Solve[%, Dt[y, x]]`

Out[37]= $\left\{\left\{\text{Dt}[y,\,x]\rightarrow\frac{-2\,x-x^2-y^2}{x^2+2\,y+y^2}\right\}\right\}$

Ecuaciones diferenciales ordinarias

La función **DSolve** nos permite resolver ecuaciones diferenciales ordinarias de orden n, expresando la solución general en función de constantes arbitrarias denominadas C[1], C[2],…, C[n], su sintaxis es:

DSolve[ecu,y[x],x]

```
In[4]:= ecu = x y'[x] + y[x] == y[x]^2 Log[x]

Out[4]= y[x] + x y'[x] == Log[x] y[x]^2

In[5]:= DSolve[ecu, y[x], x]

Out[5]= {{y[x] -> 1/(1 + x C[1] + Log[x])}}
```

```
In[8]:= DSolve[2 y[x] + y'[x] == x^2 + 2 x, y[x], x]

Out[8]= {{y[x] -> 1/4 (-1 + 2 x + 2 x^2) + e^(-2 x) C[1]}}
```

```
In[3]:= DSolve[y''[x] - 2 y'[x] + 2 y[x] == Exp[x] Sin[x], y[x], x]

Out[3]= {{y[x] -> e^x C[2] Cos[x] + e^x C[1] Sin[x] - 1/4 e^x Cos[x] (2 x + 2 Cos[x] Sin[x] - Sin[2 x])}}
```

Con la función **DSolve** podemos resolver también problemas de ecuaciones diferenciales de orden n con valor inicial.

DSolve[{ecu,y[x0]==c1,y’[x0]==c2,…,y$^{(n-1)}$ [x0]==cn}, y[x],x]

```
In[1]:= ecu = y'''[x] + y'[x] == Sin[x] + 2

Out[1]= y'[x] + y^(3)[x] == 2 + Sin[x]

In[2]:= DSolve[{ecu, y[0] == 0, y'[0] == 0, y''[0] == 2}, y[x], x]

Out[2]= {{y[x] -> 1/2 (6 + 4 x - 6 Cos[x] - 4 Sin[x] - x Sin[x])}}
```

```
In[4]:= DSolve[{(x - 1) y'''[x] - y''[x] == 0, y[2] == 2, y'[2] == 1, y''[2] == 2}, y[x], x]

Out[4]= {{y[x] -> 1/3 (4 + 3 x - 3 x^2 + x^3)}}
```

Sistemas de ecuaciones diferenciales ordinarias de primer orden

La función **DSolve** también nos permite resolver sistemas de ecuaciones diferenciales ordinarias de primer orden, expresando la solución general en función de constantes arbitrarias denominadas C[1], C[2],…, C[n], su sintaxis es:

DSolve[{ecu1,ecu2,…,ecun},{x[t],y[t]},t]

```
In[5]:= ecu1 = x'[t] - 4 x[t] - y[t] == 36 t; ecu2 = y'[t] + 2 x[t] - y[t] + 2 Exp[t] == 0;

In[6]:= DSolve[{ecu1, ecu2}, {x[t], y[t]}, t]
```

Out[6]= $\{\{x[t] \to -2 e^{-t} (-1 + 2 e^{t}) (4 + e^{2t} + 12 t - e^{t} (5 + 9 t)) + e^{2t} (-1 + e^{t}) (4 e^{-t} + e^{-2t} (-19 - 36 t) + e^{-3t} (8 + 24 t)) + e^{2t} (-1 + 2 e^{t}) C[1] + e^{2t} (-1 + e^{t}) C[2],$
$y[t] \to 4 e^{-t} (-1 + e^{t}) (4 + e^{2t} + 12 t - e^{t} (5 + 9 t)) - e^{2t} (-2 + e^{t}) (4 e^{-t} + e^{-2t} (-19 - 36 t) + e^{-3t} (8 + 24 t)) - 2 e^{2t} (-1 + e^{t}) C[1] - e^{2t} (-2 + e^{t}) C[2]\}\}$

```
In[7]:= Simplify[%]
```

Out[7]= $\{\{x[t] \to 1 - e^{t} - 6 t - e^{2t} (C[1] + C[2]) + e^{3t} (2 C[1] + C[2]),$
$y[t] \to 3 e^{t} - 2 (5 + 6 t) + 2 e^{2t} (C[1] + C[2]) - e^{3t} (2 C[1] + C[2])\}\}$

Para problemas de valor inicial para sistemas de ecuaciones diferenciales ordinarias de primer orden, la sintaxis es:

DSolve[Join[{ecu1,ecu2,…,ecun,y[t0]==c1,x[t0]==c2}], {x[t],y[t]},t]

```
In[8]:= ecu1 = x'[t] == 4 x[t] + y[t] - 36 t; ecu2 = y'[t] == -2 x[t] + y[t] - 2 Exp[t];

In[9]:= DSolve[Join[{ecu1, ecu2, x[0] == 0, y[0] == 1}], {x[t], y[t]}, t]
```

Out[9]= $\{\{x[t] \to -1 - e^{t} + 10 e^{2t} - 8 e^{3t} + 6 t,\ y[t] \to 10 + 3 e^{t} - 20 e^{2t} + 8 e^{3t} + 12 t\}\}$

3.- CÁLCULO INTEGRAL

Integral indefinida

Para hallar la primitiva de una función se utiliza el comando **Integrate[f,var]** o se puede utilizar el correspondiente símbolo que aparece en la paleta BasicInput.

In[1]:= $\text{Integrate}\left[\frac{1}{x^3 \sqrt[5]{1+\frac{1}{x}}}, x\right]$

Out[1]= $\frac{5 \left(1+\frac{1}{x}\right)^{4/5} (-4+5x)}{36 x}$

In[2]:= $\int \frac{1}{x^3 \sqrt[5]{1+\frac{1}{x}}}\, dx$

Out[2]= $\frac{5 \left(1+\frac{1}{x}\right)^{4/5} (-4+5x)}{36 x}$

Integral definida

Utilizando el comando anterior con la sintaxis, **Integrate[f,{x,a,b}]** o el correspondiente símbolo de la paleta BasicInput se calcula el valor de la integral definida.

In[3]:= $\text{Integrate}\left[\dfrac{1}{x^3\sqrt[5]{1+\frac{1}{x}}}, \{x, 3, 5\}\right]$

Out[3]= $-\dfrac{55}{27\ 2^{2/5}\ 3^{4/5}}+\dfrac{7}{2\ 5^{4/5}\ 6^{1/5}}$

In[5]:= `N[%3]`

Out[5]= 0.0338877

In[6]:= $\displaystyle\int_3^5 \frac{1}{x^3\sqrt[5]{1+\frac{1}{x}}}\,dx$

Out[6]= $-\dfrac{55}{27\ 2^{2/5}\ 3^{4/5}}+\dfrac{7}{2\ 5^{4/5}\ 6^{1/5}}$

In[7]:= `N[%]`

Out[7]= 0.0338877

Aplicaciones geométricas de la integral definida

A través del comando **Integrate** se resuelven problemas de cálculo de áreas, longitudes de curvas y volúmenes.

EJERCICIO 1

Calcular el área entre las curvas $x^2+y^2=16$; $x^2=12(y-1)$

Mediante el comando **Plot** representamos las dos funciones en el plano.

```
In[1]:= << Graphics`ImplicitPlot`

In[3]:= ImplicitPlot[{x^2 + y^2 == 16, x^2 == 12 (y - 1)}, {x, -5, 5},
          PlotStyle -> {{RGBColor[1, 0, 0], Thickness[0.01]}, {RGBColor[0, 1, 0], Thickness[0.01]}}];
```



A continuación, buscamos los puntos de corte de las dos curvas para determinar el recinto del área que se debe calcular.

In[5]:= Solve$\left[\left\{x^2+y^2==16,\ x^2==12\,(y-1)\right\},\ \{x,\ y\}\right]$

Out[5]= $\left\{\left\{y\to -14,\ x\to -6\,i\sqrt{5}\right\},\ \left\{y\to -14,\ x\to 6\,i\sqrt{5}\right\},\ \left\{y\to 2,\ x\to -2\sqrt{3}\right\},\ \left\{y\to 2,\ x\to 2\sqrt{3}\right\}\right\}$

Los puntos de corte corresponden a los valores de abscisas $x=-2\sqrt{3}$; $x=2\sqrt{3}$.

In[6]:= $\int_0^{2\sqrt{3}}\left(\sqrt{16-x^2}-\frac{x^2}{12}-1\right)dx$

Out[6]= $-\frac{2}{3}\left(\sqrt{3}-4\pi\right)$

El área entre las curvas $x^2+y^2=16$; $x^2=12(y-1)$ es por tanto:

$$A=2\left[-\frac{4}{3}(\sqrt{3}-4\pi)\right]=\frac{16\pi-4\sqrt{3}}{3}\quad u^2$$

EJERCICIO 2

Calcular el volumen del sólido obtenido al girar la región $y\le\sqrt{x}$, $y\ge x^2$, alrededor del eje de abscisas.

Mediante el comando **Plot** representamos las dos funciones en el plano. Después buscamos los puntos de corte de las dos curvas y a continuación, mediante la fórmula del volumen del sólido obtenido al girar una región alrededor del eje de abscisas, obtenemos la solución del ejercicio.

In[43]:= ImplicitPlot$\left[\left\{y^2==x,\ y==x^2\right\},\ \{x,\ 0,\ 1.1\},\right.$
PlotStyle $\to$ {{RGBColor[1, 0, 0], Thickness[0.01]}, {RGBColor[0, 1, 0], Thickness[0.01]}}];



In[44]:= Solve$\left[\left\{y^2==x,\ y==x^2\right\}\right]$

Out[44]= $\left\{\{x\to 0,\ y\to 0\},\ \{x\to 1,\ y\to 1\},\ \left\{x\to -(-1)^{1/3},\ y\to (-1)^{2/3}\right\},\ \left\{x\to (-1)^{2/3},\ y\to -(-1)^{1/3}\right\}\right\}$

Los puntos de corte corresponden a los valores de abscisas $x=0$; $x=1$.

In[45]:= $\pi\left(\int_0^1 (x-x^4)\,dx\right)$

Out[45]= $\frac{3\pi}{10}$

El volumen del sólido obtenido al girar la región alrededor del eje de abscisas es:

$$V=\frac{3\pi}{10}\quad u^3$$

EJERCICIO 3

Hallar la longitud del arco de la parábola $y=x^2+2x-3$, entre los puntos de corte con el eje de abscisas.

Representamos la función en el plano y calculamos los puntos de corte con el eje OX

In[46]:= **Solve**$[x^2 + 2x - 3 == 0, x]$

Out[46]= $\{\{x \to -3\}, \{x \to 1\}\}$

In[48]:= **Plot**$[x^2 + 2x - 3, \{x, -3, 2\}]$;



Aplicando la fórmula de la longitud del arco de una curva, $L=\int_a^b \sqrt{1+y'^2}\,dx$, junto con el comando **D** para hallar la derivada obtenemos:

In[49]:= $\int_{-3}^{1}\sqrt{1+(\partial_x x^2+2x-3)^2}\,dx$

Out[49]= $\frac{1}{8}\left(\sqrt{2}+15\sqrt{226}+\text{ArcSinh}[1]+\text{ArcSinh}[15]\right)$

In[50]:= **N[%]**

Out[50]= 28.8997

La longitud del arco de la parábola $y=x^2+2x-3$, entre los puntos de corte con el eje de abscisas, es: $L=28.89\quad u$

Integrales dobles y triples

De manera análoga se calculan integrales dobles y triples.

Integrate[f,{x,x1,x2},{y,y1,y2}]

In[52]:= $\int_0^5 \int_{-\frac{y+5}{5}}^{\frac{y+5}{5}} 1\, dx\, dy + \int_5^9 \int_{-\sqrt{9-y}}^{\sqrt{9-y}} 1\, dx\, dy$

Out[52]= $\frac{77}{3}$

In[53]:= Integrate$\left[1, \{y, 0, 5\}, \left\{x, -\frac{y+5}{5}, \frac{y+5}{5}\right\}\right]$ + Integrate$\left[1, \{y, 5, 9\}, \left\{x, -\sqrt{9-y}, \sqrt{9-y}\right\}\right]$

Out[53]= $\frac{77}{3}$

Integrales Impropias

Con un procedimiento similar al resto de integrales resueltas en los apartados anteriores, se determina la convergencia de las integrales impropias.

Integral impropia convergente

In[54]:= $\int_0^{\frac{\pi}{2}} \frac{\text{Cos}[x]}{\sqrt{1-\text{Sin}[x]}}\, dx$

Out[54]= 2

Integral impropia divergente

In[55]:= $\int_1^{\infty} \frac{x^2}{4x^2+25}\, dx$

Integrate::idiv : Integral of $\frac{x^2}{25+4x^2}$ does not converge on $\{1, \infty\}$. More…

Out[55]= $\int_1^{\infty} \frac{x^2}{25+4x^2}\, dx$

4.- SERIES NUMÉRICAS

Para calcular la suma parcial de los k primeros términos de una serie se utiliza el comando:

Sum[f,{n,1,k}]

La suma de términos consecutivos se obtiene realizando:

Sum[f,{n,h,k}]

Si la serie es convergente, su suma se obtiene con el comando:

Sum[f,{n,1,Infinity}]

Como alternativa a este commando **Sum** se puede utilizar el símbolo correspondiente de la paleta BasicInput.

EJERCICIO 4

Hallar la suma de los 5 primeros términos de la serie de término general $a_n = \frac{1}{(2n-1)^2}$

In[2]:= $\text{Sum}\left[\frac{1}{(2n-1)^2}, \{n, 1, 5\}\right]$

Out[2]= $\frac{117469}{99225}$

In[3]:= $\sum_{n=1}^{5} \frac{1}{(2n-1)^2}$

Out[3]= $\frac{117469}{99225}$

EJERCICIO 5

Sumar, si son convergentes, las serie de términos generales: $a_n = \dfrac{1}{(2n-1)^2}$; $b_n = \dfrac{n-3}{2n-1}$

In[4]:= $\text{Sum}\left[\frac{1}{(2n-1)^2}, \{n, 1, \infty\}\right]$

Out[4]= $\frac{\pi^2}{8}$

In[5]:= $\sum_{n=1}^{\infty} \frac{1}{(2n-1)^2}$

Out[5]= $\frac{\pi^2}{8}$

In[6]:= $\text{Sum}\left[\frac{n-3}{2n-1}, \{n, 1, \infty\}\right]$

```
Sum::div :  Sum does not converge. More…
Sum::div :  Sum does not converge. More…
```

Out[6]= $\sum_{n=1}^{\infty} \frac{-3+n}{-1+2n}$

El comando **Sum** permite hallar expresiones ya conocidas, como la suma de los cuadrados o los cubos de los n primeros números naturales.

In[7]:= $\sum_{p=1}^{n} p^2$

Out[7]= $\frac{1}{6} n (1+n) (1+2n)$

In[8]:= $\sum_{p=1}^{n} p^3$

Out[8]= $\frac{1}{4} n^2 (1+n)^2$

Además este comando se puede utilizar para hallar la suma de series dobles.

Sum[f(i,j),{i,1,k},{j,1,h}]

In[9]:= `Sum[i + 2 j, {i, 1, 6}, {j, 1, 10}]`

Out[9]= 870

In[11]:= $\sum_{i=1}^{6}\sum_{j=1}^{10}(i+2j)$

Out[11]= 870

5.- FÓRMULA DE TAYLOR

La ***fórmula de Taylor*** de la función $f(x)$ en el punto $x=a$ es:

$$f(x)=f(a)+\frac{f'(a)}{1!}(x-a)+\frac{f''(a)}{2!}(x-a)^2+\cdots+\frac{f^{(n)}(a)}{n!}(x-a)^n+R_{n+1}(x)$$

Si $a=0$, el desarrollo anterior recibe el nombre de ***fórmula de Maclaurin:***

$$f(x)=f(0)+\frac{f'(0)}{1!}x+\frac{f''(0)}{2!}x^2+\cdots+\frac{f^{(n)}(0)}{n!}x^n+R_{n+1}(x)$$

siendo:

$$R_{n+1}(x)=\frac{f^{(n+1)}(\theta x)}{(n+1)!}x^{n+1}\qquad \text{con}\quad \theta\in(0,1)$$

Esta fórmula de Taylor se obtiene a partir del comando **Series**.

Series[f,{x,a,n}]

In[1]:= `Series[`e^x`, {x, 2, 5}]`

Out[1]= $e^2+e^2(x-2)+\frac{1}{2}e^2(x-2)^2+\frac{1}{6}e^2(x-2)^3+\frac{1}{24}e^2(x-2)^4+\frac{1}{120}e^2(x-2)^5+O[x-2]^6$

In[2]:= `Series[Sin[x], {x, 0, 5}]`

Out[2]= $x-\frac{x^3}{6}+\frac{x^5}{120}+O[x]^6$

La función **SeriesCoefficient** de sintaxis:

SeriesCoefficient[expresión,n]

Nos determina el término de grado n en la serie de potencias representada por *expresión.*

```
In[3]:= ap = Series[(1+x)^35, {x, 0, 10}]
```

$$1 + 35x + 595x^2 + 6545x^3 + 52360x^4 + 324632x^5 + 1623160x^6 + 6724520x^7 + 23535820x^8 + 70607460x^9 + 183579396x^{10} + O[x]^{11}$$

```
In[4]:= SeriesCoefficient[ap, 7]
Out[4]= 6724520
In[5]:= SeriesCoefficient[Series[(1+x)^35, {x, 0, 10}], 9]
Out[5]= 70607460
```

A partir de la fórmula de Taylor, se puede obtener el correspondiente polinomio de Taylor utilizando el comando **Normal**, de sintaxis:

Normal[expresión]

```
In[6]:= Series[e^x, {x, 2, 6}]
```

$$e^2 + e^2 (x-2) + \frac{1}{2} e^2 (x-2)^2 + \frac{1}{6} e^2 (x-2)^3 + \frac{1}{24} e^2 (x-2)^4 + \frac{1}{120} e^2 (x-2)^5 + \frac{1}{720} e^2 (x-2)^6 + O[x-2]^7$$

```
In[7]:= Normal[%]
```

$$e^2 + e^2 (-2+x) + \frac{1}{2} e^2 (-2+x)^2 + \frac{1}{6} e^2 (-2+x)^3 + \frac{1}{24} e^2 (-2+x)^4 + \frac{1}{120} e^2 (-2+x)^5 + \frac{1}{720} e^2 (-2+x)^6$$

```
In[8]:= Normal[Series[e^x, {x, 2, 6}]]
```

$$e^2 + e^2 (-2+x) + \frac{1}{2} e^2 (-2+x)^2 + \frac{1}{6} e^2 (-2+x)^3 + \frac{1}{24} e^2 (-2+x)^4 + \frac{1}{120} e^2 (-2+x)^5 + \frac{1}{720} e^2 (-2+x)^6$$

Para obtener los polinomios de Taylor de una función de grado menor o igual que p, introduciremos el comando:

Table[Normal[Series[f,{x,a,n}]],{n,1,p}]

Estos polinomios se pueden representar de firma simultánea a través de los comandos **Plot** y **Evaluate**.

```
In[9]:= Table[Normal[Series[e^x, {x, 0, n}]], {n, 1, 4}]
```

$$\left\{1 + x,\ 1 + x + \frac{x^2}{2},\ 1 + x + \frac{x^2}{2} + \frac{x^3}{6},\ 1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24}\right\}$$

```
In[10]:= Plot[Evaluate[%], {x, -2, 2}];
```



6.- TRANSFORMADA DE LAPLACE

La función **LaplaceTransform** nos permite determinar la transformada de Laplace de una función $f(t)$ respecto del parámetro s.

LaplaceTransform[f[t],t,s]

```
In[11]:= LaplaceTransform[1, t, s]
```

Out[11]= $\frac{1}{s}$

```
In[13]:= LaplaceTransform[Sin[b t], t, s]
```

Out[13]= $\frac{b}{b^2 + s^2}$

```
In[14]:= LaplaceTransform[Exp[-a t] Sin[b t], t, s]
```

Out[14]= $\frac{b}{b^2 + (a+s)^2}$

In[17]:= LaplaceTransform$\left[\frac{(1 - \mathrm{Cos}[a\,t])}{a^2}, t, s\right]$

Out[17]= $\frac{\frac{1}{s} - \frac{s}{a^2+s^2}}{a^2}$

```
In[18]:= Simplify[%]
```

Out[18]= $\frac{1}{a^2 s + s^3}$

La función **InverseLaplaceTransform** nos permite determinar la transformada inversa de Laplace de una función $F(s)$ en función de la variable t.

InverseLaplaceTransform[F[s],s,t]

In[19]:= InverseLaplaceTransform$\left[\frac{s}{s^2+b^2}, s, t\right]$

```
Out[19]= Cos[b t]
```

In[20]:= InverseLaplaceTransform$\left[\frac{1}{s+a}, s, t\right]$

Out[20]= $e^{-a t}$

In[21]:= InverseLaplaceTransform$\left[\frac{1}{(s-b)\,(s-a)}, s, t\right]$

Out[21]= $\frac{e^{a t} - e^{b t}}{a - b}$

EJERCICIO 6

Obtener las funciones generatrices de las transformadas:

$$F(s) = \frac{s^2 - 2s + 3}{s^3 - s^2 - s + 1} \qquad G(s) = \frac{3s+7}{s^2 - 2s - 3}$$

```
In[22]:= InverseLaplaceTransform[(s^2 - 2 s + 3)/(s^3 - s^2 - s + 1), s, t]
```

$$\text{Out[22]}= \frac{3\,e^{-t}}{2} - \frac{e^{t}}{2} + e^{t}\,t$$

```
In[24]:= InverseLaplaceTransform[(3 s + 7)/(s^2 - 2 s - 3), s, t]
```

$$\text{Out[24]}= -e^{-t} + 4\,e^{3t}$$

La transformada de Laplace puede utilizarse para resolver ecuaciones y sistemas de ecuaciones diferenciales lineales.

EJERCICIO 7

Sea la ecuación diferencial lineal de orden 3 con coeficientes constantes definida por:

$$y''' - 3y'' + 3y' - y = x^2 e^x; \qquad y(0) = 1,\ y'(0) = 0,\ y''(0) = -2$$

Determinar la solución de la ecuación diferencial.

```
In[25]:= ecu = y'''[t] - 3 y''[t] + 3 y'[t] - y[t] == t^2 Exp[t]
```

$$\text{Out[25]}= -y[t] + 3\,y'[t] - 3\,y''[t] + y^{(3)}[t] == e^{t}\,t^{2}$$

```
In[26]:= transecu = LaplaceTransform[ecu, t, s] /. {y[0] → 1, y'[0] → 0, y''[0] → -2}
```

$$\text{Out[26]}= 2 - s^2 - \text{LaplaceTransform}[y[t], t, s] + s^3\,\text{LaplaceTransform}[y[t], t, s] + 3\,(-1 + s\,\text{LaplaceTransform}[y[t], t, s]) - 3\,(-s + s^2\,\text{LaplaceTransform}[y[t], t, s]) == \frac{2}{(-1+s)^3}$$

```
In[27]:= sol = Solve[transecu, LaplaceTransform[y[t], t, s]]
```

$$\text{Out[27]}= \left\{\left\{\text{LaplaceTransform}[y[t], t, s] \to \frac{1 + 6s - 13s^2 + 13s^3 - 6s^4 + s^5}{(-1+s)^6}\right\}\right\}$$

```
In[28]:= InverseLaplaceTransform[(1 + 6 s - 13 s^2 + 13 s^3 - 6 s^4 + s^5)/(-1 + s)^6, s, t]
```

$$\text{Out[28]}= \frac{1}{60}\,e^{t}\,(60 - 60\,t - 30\,t^2 + t^5)$$

La solución de la EDO es:

$$y = \frac{1}{60} e^{x} (60 - 60x - 30x^2 + x^5)$$

EJERCICIO 8

Determinar la solución de la ecuación diferencial de condiciones iniciales con valores nulos:

$$x'' + 2ax' + a^2 x = t^2 e^{-at}$$

```
In[33]:= ecu = x''[t] + 2 a x'[t] + a^2 x[t] == t^2 Exp[-a t]
Out[33]= a^2 x[t] + 2 a x'[t] + x''[t] == e^(-a t) t^2

In[34]:= transecu = LaplaceTransform[ecu, t, s] /. {x[0] → 0, x'[0] → 0}
Out[34]= a^2 LaplaceTransform[x[t], t, s] +
          2 a s LaplaceTransform[x[t], t, s] + s^2 LaplaceTransform[x[t], t, s] == 2/(a + s)^3

In[35]:= sol = Solve[transecu, LaplaceTransform[x[t], t, s]]
Out[35]= {{LaplaceTransform[x[t], t, s] → 2/(a + s)^5}}

In[36]:= InverseLaplaceTransform[2/(a + s)^5, s, t]
Out[36]= 1/12 e^(-a t) t^4
```

EJERCICIO 9

Determinar la solución particular del sistema de ecuaciones diferenciales que verifica la condición $x(0) = 1,\ y(0) = 0$.

$$\begin{cases} x' = (a+1)x - y \\ y' = x + (a-1)y \end{cases}$$

```
In[40]:= ecu = {x'[t] == (a + 1) x[t] - y[t], y'[t] == x[t] + (a - 1) y[t]};

In[41]:= transecu = LaplaceTransform[ecu, t, s] /. {x[0] → 1, y[0] → 0}
Out[41]= {-1 + s LaplaceTransform[x[t], t, s] ==
            (1 + a) LaplaceTransform[x[t], t, s] - LaplaceTransform[y[t], t, s],
          s LaplaceTransform[y[t], t, s] ==
            LaplaceTransform[x[t], t, s] + (-1 + a) LaplaceTransform[y[t], t, s]}

In[42]:= sol = Solve[transecu, {LaplaceTransform[x[t], t, s], LaplaceTransform[y[t], t, s]}]
Out[42]= {{LaplaceTransform[x[t], t, s] → -(-1 + a - s)/(a - s)^2, LaplaceTransform[y[t], t, s] → 1/(a - s)^2}}

In[44]:= InverseLaplaceTransform[-(-1 + a - s)/(a - s)^2, s, t]
Out[44]= e^(a t) (1 + t)

In[45]:= InverseLaplaceTransform[1/(a - s)^2, s, t]
Out[45]= e^(a t) t
```

La solución del sistema de ecuaciones diferenciales lineales es:

$$x(t) = e^{at}(1+t) \qquad y(t) = t\,e^{at}$$